
              TI-84 PLUS CE AND THE TI-INNOVATOR™ ROVER						       			                    TEACHER NOTES
	Overview:
	Goals:

	In this activity students will explore inequalities and the number line while writing code to navigate a set of challenges. Students will apply their knowledge of compound inequalities to write programs for Rover to demonstrate the inequalities on the number line. This activity is appropriate for students familiar with graphing inequalities on a number line diagram.
	Students will:
· use their knowledge of compound inequalities and graphs of compound inequalities to create and edit TI-BASIC programs that include several commonly used Rover and calculator commands.


	Command Support: 

		Command
	Example
	Behavior

	CONNECT RV
	Send("CONNECT RV")
	Associates the Rover with the TI-Innovator Hub.

	RV FORWARD
	Send("RV FORWARD 10")
	Rover drives forward 10 units. The default speed is 2 units per second.

	RV BACKWARD
	Send("RV BACKWARD 10")
	Rover drives backward 10 units. The default speed is 2 units per second.

	RV STOP 
	Send("RV STOP")
	Rover stops. This command is executed as soon as Rover receives it.

	SET RV.COLOR <r g b>
	Send("SET COLOR 255 0 0")
	Turns the color LED on with the color red.

	Wait <time>
	Wait 3
	The calculator will wait 3 seconds before moving to the next line.

	getKey
	getKey→K
	getKey is a function that returns a value associated with the last key pressed while a programming is running. The keypresses are mapped to values that represent rows and columns. The clear key is in row 4 column 5 and so pressing clear returns 45. In this program the value of getKey is stored to the variable K. 

	READ RV.WAYPOINT.X
	Send("READ RV.WAYPOINT.X")
	Reads the current x position in Rover units. 

	Get(<variable>)
	Get(X)
	Stores the waypoint measurement into a variable named X. The value stored will contain the measurement from the immediately preceding READ command. Note: a Get( command must immediately follow a READ command.

	Output(<row 1-10>, <column 1-26>,<expression> <variable name> or <string>)
Note: It is recommended to use the ClrHome command before using Output commands
	Output(4,1,"POSITION=  ")
Output(4,10,X)
	When variable X has a value of 7.6, the following line is displayed on the calculator:  
POSITION= 7.6






	Command Support continued: 

		Command
	Example
	Behavior

	While <Boolean expression>
<statements>
End
	0→K 
While K=0  
 Send("READ RV.WAYPOINT.X")
 Get(X)
 Output(3,1"X POSITION= ")
 Output(3,13,X)
 getKey→K
End
	The variable K is set to an initial value of 0 using the Store function, →. The statements in the While loop are executed until the escape key is pressed. The While loop continues as long as the Boolean expression evaluates to “true”. Each time through the loop the x position is read by the READ RV.WAYPOINT.X command and stored in the variable X and the last key pressed is stored in the variable K 

	If <Boolean expression> Then 
   <statements>
End
	If X<2 
Then
 Send("SET COLOR 255 0 0")
End
	Checks to determine if the value of variable X is less than 2. If the statement is true then the block of commands in the If.. End block are executed. Otherwise the block is skipped. In this case, when the position is less than 2 the calculator will send a command to the TI-Innovator to set the COLOR to be red.

	<Boolean expression> and <Boolean expression>
	If X≥2 and X<4 
Then
 Send("SET COLOR 0 255 0")
End
	When both expressions are true the “and” function is “true” and the statements are executed. Otherwise, the function returns false and the statements are skipped.


See TI-Innovator Technology eGuide for more background https://education.ti.com/html/webhelp/EG_Innovator/EN/index.html.






	Setup:	
	Materials:

	Students may work in groups of two or three. An example number line layout is below. Use this for the position of Rover. Rover’s position corresponds to the center of the axle as shown by the red segment below. 1 Rover unit is 10 cm. Mark with 10 cm units to match with Rover units, from -10 units to +10 units. 

	· TI-84 Plus CE graphing calculator 
· Calculator unit-to-unit cable (USB mini A to USB mini B cable)
· TI-Innovator™ Hub
· TI-Innovator™ Rover
· Number line for position. Use a 2 meter course, from -10 units to +10 units. 1 Rover unit is 10 cm. 
· Student “Challenge” cards have been provided and can be printed/ cut into individual task cards, and distributed to students (optional)

	Student Activity:
	Teacher Notes:

	Sit in small groups with your calculator and supplies for this activity. Practice the guidance modeled by your teacher.

	Review and introduce the calculator, Hub, and Rover commands needed for this activity. In preparation for the coding on this activity, refer to Unit 1 Skill Builder 1 of the “10 Minutes of Code” activities and Unit 1 Skill Builder 2 and Unit 4 Skill Builder 1 and 2 of the “10 Minutes of Code” with TI-Innovator activities at https://education.ti.com/en/activities/ti-codes
· Attach Rover




	Challenge 1: Use the SET RV.COLOR command to explore using the color LED. Try to find RGB values for the primary and secondary colors.

e.g. Send("SET RV.COLOR 255 155 0") will make yellow.
 
	Teacher Guidance during Challenge 1:
· Students may have different values for mixing colors. In this activity we will primarily use the red, green and blue components individually but this challenge allows them a chance to explore settings for other colors.

· Example Program:
PROGRAM:COLOR
:Send("CONNECT RV") 
:Send("SET RV.COLOR 255 155 0") 


	Challenge 2: Use the Output( command to display your name at several locations on the screen.



	Teacher Guidance during Challenge 2:
· Use the Output( command (Program I/O menu). Output takes three inputs (also known as “arguments”. The first input is the row number to display on (1-10). The second input is the column to display the first character (1-26). The third is the text string, value, variable, etc. to display. 
· ClrHome is recommended to clear the home screen for display (Program I/O menu). 

· Example program:
PROGRAM:NAME
:ClrHome
:Output(1,1, "HELLO MY NAME IS")
:Output(2,1, "ROVER")




	Challenge 3: Have Rover drive 5 units forward. Use the READ RV.WAYPOINT.X to read and display Rover’s horizontal position when Rover is finished driving. 


	Teacher Guidance during Challenge 3:
· READ RV.WAYPOINT.X reads Rover’s current X position in Rover units, relative to the origin established by CONNECT RV. The READ RV.WAYPOINT.X command is found under the READ RV PATH menu.
· Note: When CONNECT RV is executed, along with establishing communication between the calculator and Rover, it also sets the current position to (0,0) and the heading to 0 degrees (pointing down the positive x-axis). For this activity we are ignoring its y position.
· It will be helpful for students to think about how long it will take Rover to drive forward 5 units and use a wait command to wait until Rover has stopped driving before reading Rover’s position. 
· Note: Rover’s speed default speed is 2 units per second. Students will need to make note of this in order to determine how long to wait. 
· Students may also notice a roughly 1 second delay between when the program is executed and Rover starts. Students should account for this time in their wait command before reading the Rover’s position. 
· Have students note there may be some variability in the position that is measured.
· Example Program:
PROGRAM:POSITION
:ClrHome
:Send("CONNECT RV")
:Send("RV FORWARD 5")
:Wait 3.5
:Send("READ RV.WAYPOINT.X")
:Get(X)
:Output(5,1,"POSITION=")
:Output(5,10,X)
· Discussion Starters
· The following are suggested discussion starters to engage your students with the mathematics inherent in Challenge 3: 
· How long did you wait to read the position of Rover? Explain how you determined how long to wait before reading the position of Rover? 
· If you changed Rover’s speed to 1.5 units per second, explain how this would change the amount of time you would need to wait before reading the distance Rover traveled to be sure it was finished driving. Write and solve an equation for the wait time.



	Challenge 4: Use a While..End loop to flash the LED red, then green, then blue, each for 1 second until a key is pressed. 


	Teacher Guidance during Challenge 4:
· Loops are used to repeat a set of commands. A While loop repeats a set of commands as long as a specified condition is “true”. In TI-BASIC the While loop is defined using a Boolean operator. 
· This While loop in will execute as long as the value of getKey is equal to 0 showing no key has been pressed. As soon as any key is pressed, the value of K will change and the loop will be skipped over and the program will execute the commands after the While loop. 
· Prior to the loop, the value of K is set to an empty string so the While loop will execute at least once. 
· Students will need to remember to press a key to exit the program. 
· The equals sign is under the TEST menu, found by pressing y» and choosing '=' equals. 

· Example Program:
PROGRAM:COLORS
: 0→K
:While K=0
:Send("SET RV.COLOR 255 0 0")
:Wait 1
:Send("SET RV.COLOR 0 255 0")
:Wait 1
:Send("SET RV.COLOR 0 0 255")
:Wait 1
:getKey→K
:End
:Send("SET RV.COLOR 0 0 0")
EndPrgm





	Challenge 5: Have Rover drive 5 units forward. Predict the amount of time for Rover to reach 4 units, and Read Rover’s position at that time. If the value returned is equal to 4 turn the LED green, if the value returned is less than 4 turn the LED red, and if the value is greater than 4, turn the LED blue. 







	Teacher Guidance during Challenge 5:
· This challenge will require students to edit their previous program and add in the conditional statements to control the RGB LED.
· The inequalities are under the TEST menu, found by pressing y». The If..Then..End blocks will only evaluate the contained commands if the condition given after If is “true”. If the inequality statement is “false”, then the block of commands is skipped.
· Example Program:
PROGRAM:POSITN2
:Send("CONNECT RV")
:Send("RV FORWARD 5")
:Wait 3
:Send("READ RV.WAYPOINT.X")
:Get(X)
:If X=4 
:Then
:Send("SET RV.COLOR 0 255 0")
:End
:If X<4
:Then
:Send("SET RV.COLOR 255 0 0")
:End
:If X>4
:Then
:Send("SET RV.COLOR 0 0 255")
:End
· Discussion Starters
· The following are suggested discussion starters to engage your students with the mathematics inherent in Challenge 5:
· If your LED turns blue, what adjustment can you make to your program to make it turn green or red? Explain. 
· If Binu’s LED is red after his drive and Jisha’s LED is blue, what can you say about the time when Rover is at 4 units from zero? Explain. 



	Challenge 6: Have Rover drive on the number line between 0 and 10. While Rover is driving, read its position and control the LED so that the LED displays colors corresponding to the number line diagram below.

	Teacher Guidance during Challenge 6:
· For this challenge, students will benefit from going through the discussion starters before attempting to write the code.
· This challenge will require students putting Rover in motion using RV FORWARD and RV BACKWARD commands and then, while rover is moving, students use the While loop to monitor Rover’s position while it drives. 
· The While loop should have a Boolean expression looking for a key press similar to challenge 4. 
· If students decide to drive to numbers less than zero or greater than 10 units, they will need to include If..Then.. conditions for those points as well. The example program does not address driving outside of the domain 0 to 10. 
· Students may want to use compound inequalities of the form 5<x<10 for the If..Then.. conditions, however, this will not work on the TI-84 Plus CE. However, students will benefit from using the “and” operator and considering the logic inherent in the statement 5<x<10. This is equivalent to 5<x and x<10.  The "and" operator is under the test menu, found by pressing y», then choose LOGIC. Also under the test menu, there is also a menu showing CONDITIONS which has the structure already set up. Students may take advantage of this menu. 
· Students will need to remember to press any key to stop the program. 
· Example Program:
PROGRAM:INEQUAL1
:Send("CONNECT RV")
:0→X
:0→K
:Send("SET RV.COLOR 0 0 255")
:Send("RV FORWARD 10")
:Send("RV BACKWARD 10")
:While K=0
:Send("READ RV.WAYPOINT.X")
:Get(X)
:If 0≤X and X<2 
:Then
:Send("SET RV.COLOR 0 0 255")
:End
:If 2≤X and X<5
:Then
:Send("SET RV.COLOR 255 0 0")
:End
:If 5≤X and X<10
:Then
:Send("SET RV.COLOR 0 255 0")
:End
:getKey→K
:End
:Send("SET RV.COLOR 0 0 0")
:Send(“RV STOP”)
· Discussion Starters
· The following are suggested discussion starters to engage your students with the mathematics inherent in Challenge 6:
· For what values of x will you turn on the RGB LED? Write an inequality to represent your answer.
· Write three inequalities defining the boundaries for each color.
· If the x position of Rover is 5, what color should the RGB LED be? Explain how you know.
· Maria says that a command to turn off the LED’s should be included for when x positon is greater than or equal to 10 units, explain why you agree or disagree with her statement.
· 
For which values of x should the LED be off? Write your answer as an inequality statement or statements and then write the If..Then.. blocks that could be added to your code to accommodate driving outside of . 
· If Rover starts at x position 0, should the LED be on? If so, what color? Explain.

Challenge 7: Have Rover drive on the number line between 0 and 10. While Rover is driving, read its position and control the LED so that the LED displays colors corresponding to the number line diagram below.

	Teacher Guidance during Challenge 7:
· For this challenge, students will benefit from going through the discussion starters before attempting to write the code.
· Note: Students could make use of the “or” operator for when the LED should be off. An alternative version of the program is listed after the discussion starters.
 
· Example Program:
PROGRAM:INEQUAL2
:Send("CONNECT RV")
:0→X
:0→K
:Send("SET RV.COLOR 0 0 0")
:Send("RV FORWARD 10")
:Send("RV BACKWARD 10")
:While K=0
:Send("READ RV.WAYPOINT.X")
:Get(X)
:If 0≤X and X<1
:Then
:Send("SET RV.COLOR 0 0 0")
:End
:If 1≤X and X<3
:Then
:Send("SET RV.COLOR 255 0 0")
:End
:If 3≤X  and X<4
:Then
:Send("SET RV.COLOR 0 0 0")
:End
:If 4≤X and X≤6
:Then
:Send("SET RV.COLOR 0 255 0")
:End
:If 6<X and X<8
:Then
:Send("SET RV.COLOR 0 255 0")
:End
:If 8≤X
:Then
:Send("SET RV.COLOR 0 0 0")
:End
:getKey→K
:End
:Send("SET RV.COLOR 0 0 0")
:Send("RV STOP")
· Discussion Starters
· The following are suggested discussion starters to engage your students with the mathematics inherent in Challenge 7:
· For what values of x will you turn on the RGB LED? Write your answer in inequality statements.
· For what values of x will the LED be turned off? Write your answer in inequality statements.
· Write inequalities defining the boundaries for each color.
· If the x position of Rover is 6, what color will the RGB LED need to be? Explain how you know.
· Maria says that a command to turn off the LED’s should be included for when x positon is greater than or equal to 8 units, explain why you agree or disagree with her statement.
· If Rover starts at x position 0, should the LED be on? If so, what color? Explain.

· Alternative example program using “or”. This example combines all the times when the LED is off into one conditional statement. 

PROGRAM:INEQUAL2
:Send("CONNECT RV")
:0→X
:0→K
:Send("SET RV.COLOR 0 0 0")
:Send("RV FORWARD 10")
:Send("RV BACKWARD 10")
:While K=0
:Send("READ RV.WAYPOINT.X")
:Get(X)
:If 0≤X<1 or 3≤X<4 or 8≤X Then
:Then
:Send("SET RV.COLOR 0 0 0")
:End
:If 1≤X and X<3
:Then
:Send("SET RV.COLOR 255 0 0")
:End
:If 4≤X and X≤6
:Then
:Send("SET RV.COLOR 0 255 0")
:End
:If 6<X and X<8
:Then
:Send("SET RV.COLOR 0 255 0")
:End
:getKey→K
:End
:Send("SET RV.COLOR 0 0 0")
:Send("RV STOP")




	Challenge 8: Have Rover drive on the number line between -10 and 10.While Rover is driving, read its position and control the LED so that the LED displays colors corresponding to the description below.


· While Rover’s position is less than or equal to zero, the LED is magenta.
· While Rover’s position is greater than 0 and less than 2, the LED is off. 
· While Rover’s position is greater than or equal to 2 and less than or equal to 4, the LED is red. 
· While Rover’s position is greater than 4 and less than 5, the LED is blue.
· While Rover’s position is greater than or equal to 5 and less than 10, the LED is green.  
· While Rover’s position is greater than or equal to 10, the LED is yellow.
	Teacher Guidance during Challenge 8:
· For this challenge it will be helpful for students to draw and color a number line diagram representing the challenge before translating the challenge into the code. 
· Students will need to use RV BACKWARD to make sure Rover’s position can be less than zero. 
· As in the previous challenges, this challenge will require students putting Rover in motion using RV FORWARD and RV BACKWARD commands and then, while Rover is moving, students use the While loop to monitor Rover’s position while it drives. 

· Example Program:
PROGRAM:VERBAL
:Send("CONNECT RV")
:0→X
:0→K
:Send("SET RV.COLOR 255 0 255")
:Send("RV BACKWARD 5")
:Send("RV FORWARD 16")
:While K=0
:Send "READ RV.WAYPOINT.X"
:Get(X)
:If X≤0 
:Then
:Send("SET RV.COLOR 255 0 255")
:End
:If 0<X and X<2 
:Then
:Send("SET RV.COLOR 0 0 0")
:End
:If 2≤X and X≤4
:Then
:Send("SET RV.COLOR 255 0 0")
:End
:If 4<X and X<5
:Then
:Send("SET RV.COLOR 0 0 255")
:End
:If 5≤X and X<10
:Then
:Send("SET RV.COLOR 0 255 0")
:End
:If 10≤X
:Then
:Send("SET RV.COLOR 255 155 0")
:End
:getKey→K
:End
:Send("SET RV.COLOR 0 0 0")
:Send("RV STOP")
· Discussion Starters
· The following are suggested discussion starters to engage your students with the mathematics inherent in Challenge 8:
· For what values of x will you turn on the RGB LED? Write your answer in inequality statements.
· For which values of x will you turn off the RGB LED? Write your answers in inequality statements. 
· Maggie made her Rover drive backward 10 units first, she says the LED should stay magenta until she drives forward 10 units and then it should turn off. Would you agree or disagree? Explain.  




©2019 Texas Instruments Incorporated	6	education.ti.com
image2.jpeg




image3.png
peG [I] X





image4.wmf
010

x

£<


oleObject1.bin

image5.png
*Moody_Driver





image6.jpeg




